Compilation of Compound Terms in Prolog

Micha Meier ECRC-ECRC-95-12 technical report ECRC-ECRC-95-12

Compilation of Compound Terms in Prolog

Micha Meier

European Computer-Industry Research Centre GmbH (Forschungszentrum)

Arabellastrasse 17 D-81925 Munich Germany Tel. +49 89 9 26 99-0 Fax. +49 89 9 26 99-170 Tlx. 52 69 10

I

cflEuropean Computer-Industry Research Centre, 1995

Although every effort has been taken to ensure the accuracy of this report, neither the authors nor the European Computer-Industry Research Centre GmbH make any warranty, express or implied, or assume any legal liability for either the contents or use to which the contents may be put, including any derived works. Permission to copy this report in whole or in part is freely given for non-proőt educational and research purposes on condition that such copies include the following:

1. a statement that the contents are the intellectual property of the European Computer-Industry Research Centre GmbH 2. this notice 3. an acknowledgement of the authors and individual contributors to this work Copying, reproducing or republishing this report by any means, whether electronic or mechanical, for any other purposes requires the express written permission of the European Computer-Industry Research Centre GmbH. Any registered trademarks used in this work are the property of their respective owners.

For more information please contact : michaecrc.de

II

Abstract

The execution of a compiled Prolog program can spend a signiőcant amount of time in the uniőcation of compound terms. We show that in the Warren Abstract Machine [7], the approach to compile this uniőcation may be unnecessarily inefőcient. When we analyse what are the redundant operations that the WAM executes, we can see that the inefőciency is mainly caused by the breadth-first approach to traverse the structures during the uniőcation. We present here a method to compile the uniőcation of compound terms which is based on a depth-first approach and show that it is both more general and more efőcient than that of the original WAM. Furthermore we present a more efőcient approach to compile compound terms in the body and also describe several possible optimizations. Our method was used in the implementation of the SEPIA system [4] developed at ECRC.

III

1 Introduction

The execution of Prolog programs consists mainly of procedure invocation and uniőcation. The uniőcation itself is estimated to account for about 70 % - 80 % of the execution time. The uniőcation of compound terms, i.e. expressions f (arg_1, . . . , argn), may consume a considerable amount of time, because during the uniőcation these compound terms have to be constructed or scanned. For example, on a set of 15 large benchmark programs measured with an emulator-based Prolog system [4], about 25 % of all emulated abstract instructions accounted for the uniőcation of compound terms.

The WAM [7] presents a complete and compact scheme to compile compound terms, both in head and subgoal arguments. We have analyzed the WAM behavior during the uniőcation of compound terms and have found that sometimes it is not efőcient enough, because redundant operations are executed. We will őrst describe the problems in the WAM and then present a different scheme which tries to avoid them. The reader is assumed to be ŕuent in the WAM, for a good introduction see e.g. [1].

2 Compound Argument Uniőcation in the WAM

In compiled Prolog, the compound arguments are handled differently in the head and in the body of a clause. A compound term in the clause body is compiled into a sequence of instructions that actually create this term on the heap and put a reference to it into an argument register. A compound term in the head is compiled into instructions which perform the uniőcation with the corresponding argument in the register. The uniőcation is a general two-way matching procedure, which can result in binding variables both in the caller and in the callee. Therefore, the uniőcation instructions must be able to both decompose the caller argument and check if it matches with the clause head, and to create the representation of a compound term on the heap in case that the corresponding slot in the caller argument is a variable. This is the so-called structure copying as opposed to structure sharing where new structures are not explicitly created, but shared instead.

2.1 Head Arguments

A compound head argument represents a tree structure which has to be traversed during the uniőcation with the caller. The main contribution of the WAM here is the introduction of the mode ŕag to which some instructions are sensitive. This makes it possible to generate only one instruction per source item and the bytecode is very compact. If the procedure argument is being decomposed in the uniőcation, that is, when its main functor is identical with the main functor of the head argument, the machine is said to be in read mode. Conversely, if the caller argument is a variable, the machine is in write mode where the structure will be constructed on the heap. The unify instructions that are emitted for the arguments of the structure depend on the current mode: in the read mode they test the type of the caller argument, in the write mode the argument is created on the heap. The arguments of a structure are uniőed in the same mode as the structure functor. When the structure is being constructed on the heap, its arguments are pushed using the heap top pointer H, when an existing structure is decomposed, the S pointer is used to point to the currently uniőed argument. For example, the head structure in

p(f(g(...), h(...), k(...))) :- : : :

is in the WAM compiled into get structure f/3, A_1 unify variable X_1 unify variable X_2 unify variable X_3 get structure g/n_1, X_1 unify : : : <arguments of g> : : :

get structure h/n_2, X_2 unify : : : <arguments of h> : : :

get structure k/n_3, X_3 unify : : : <arguments of k> : : :

If this procedure is called with ?- p(f(g(a), B, C)), the őrst and second block of unify instructions is executed in read mode because the head compound term matches the corresponding functors f/3 and g/n_1 in the caller. The remaining two blocks of unify instructions are executed in write mode, because the structures h/n_2 and k/n_3 must be constructed and bound to B and C respectively.

Let us consider this example from another point of view. We can see that the code generated by the WAM is the same as the code for (when =/2 is expanded in-line)

p(f(X1, X2, X3)) :- X1 = g(...), X2 = h(...), X3 = k(...).

where X1, X2 and X3 are temporary variables. The uniőcation of the main structure is completely separated from the uniőcation of its compound arguments. As the uniőcation of the temporaries can be written in any order, the traversal of the rest of the tree can be made in any manner, as long as this way of assigning compound arguments to temporaries is kept.

This model is very easy to understand and efőcient enough to implement on bytecode emulators, however it has several inherent problems:

fl The write mode is not propagated to the substructures. Whenever a

structure is uniőed in write mode, it is clear that all its substructures will also be uniőed in write mode. The WAM does not use this property, in a substructure in write mode the following is executed:

{ the instruction unify variable pushes a new free variable on the heap and stores a pointer to it in a temporary variable { The instruction get structure later dereferences this variable, őnds out that it is uninstantiated (although this is known at compile time), and thus it sets the write mode. The variable is then bound to a new structure on the heap and for this binding even a trail test has to be performed.

Several memory accesses, dereferencing, tag test and trail test are redundant. This is the consequence of the fact that the uniőcation of a term is completely separated from the uniőcation of its compound arguments.

fl References to all substructures are saved into temporary variables.

Although Prolog programmers do not seem to like deeply nested structures (except for tail-recursive ones like lists), often their arity might be quite high [3] and so many temporaries are necessary. This increases the number of used registers, or causes additional memory accesses for temporaries allocated in memory.

fl Every unify instruction must test the mode bit. In fact every unify

instruction is just a shorthand for two different instructions, one read and one write. In native code implementations, and especially on processors with an instruction pipeline, these frequent tests break the instruction ŕow and slow down the execution. Most commercial systems split the unify instructions explicitly to avoid the tests.

fl In a term like f(g(a), h(b), i(c)), the code starts with

get structure f/3, A_i unify variable X_1 unify variable X_2 unify variable X_3

If the caller argument is f(g(b), B, C), the uniőcation of g(a) and g(b) fails and so the two unify variable instructions were useless. The same effect could be achieved by just saving a pointer to the second argument.

fl If the information about the instantiations of the arguments is available,

e.g. in the form of mode declarations, it can be used only if the unify instructions are explicitly divided into read and write instructions. The compiler can then suppress generation of code that would never be executed.

fl This sort of breadth-őrst traversal can be always simulated using a

bounded depth-őrst traversal, however the converse is not true. Therefore the depth-őrst is more general and a better candidate for the compilation technique.

2.2 Body Arguments

In the WAM, compound terms in subgoal arguments are compiled using the same unify instructions like in the clause head. The only difference is that the structures are built bottom-up, without the get instructions, so that dereferencing and trail tests are avoided. The other problems remain, though. Although it is known that the instructions will be executed in write mode, each unify instruction tests the mode ŕag. It is also necessary to use temporary variables to build the structure. However, it is known at compile time that the structure will be created in consecutive locations at the heap top, and all pointer offsets are őxed. Therefore, no temporary variables are really necessary to create the structure.

3 Depth-First Method

Our method has the following differences compared to the WAM:

fl an explicit distinction between read and write instructions is made, the

compiler generates separate read and write mode sequences,

fl nested structures are uniőed in a depth-first manner, fl no code is duplicated, the execution can switch from the read to the

write sequence and back,

fl the body compound arguments are built top-down in a breadth-first

manner, no temporary variables are necessary.

3.1 Head Arguments

While the WAM saves all compound arguments into temporary variables, in the depth-őrst approach it is enough to save a pointer to the argument following a compound one. No temporary variable is needed for the last structure argument. Note for example, that no temporaries are necessary to unify a list containing only constants. The temporaries can be viewed as a return stack whose depth is known at compile time. When the uniőcation of a substructure is őnished, the execution proceeds with the next argument whose address is saved in the corresponding temporary variable.

Since the two sequences are now compiled separately, the get structure and get list instructions are slightly modiőed. For the write mode the execution continues in sequence, for read mode a branch to a given label is executed. At the end of the write mode sequence there is a branch instruction that causes a branch to the end of the read mode sequence.

Even when the two sequences are split, it is not possible to avoid some tests. Since part of the structure may be uniőed in read and part in write mode, branches from one sequence to the other one might be necessary (unless one wants to duplicate the code, but this is exactly what we try to avoid when compiling into native code). The basic observation is as follows: If a structure is uniőed in read mode, all its parent and sibling structures are also uniőed in read mode. Therefore, when the uniőcation of this structure is őnished, the execution continues directly in the read mode sequence. This is the upward propagation of the read mode. The compound arguments of a structure, however, might have to be uniőed in write mode if they are matched with a free variable. Hence at the beginning of the uniőcation of a structure in read mode it is necessary to perform a test and possibly branch into the write mode sequence.

If a structure is uniőed in write mode, all its substructures are also uniőed in write mode, i.e. the write mode is propagated downwards, but its parent structure might be uniőed in read mode. Therefore, at the end of each structure uniőed in write mode the system has to test if a branch back into the read mode sequence has to be made. The condition when to jump from the write mode sequence back to the read sequence is as follows: at the end of every compound term the address of the next argument to unify is obtained from a temporary variable. The tag of this variable stores a ŕag which decides whether to jump back to the read mode sequence or whether to continue with the next instruction.

In write mode the whole structure skeleton has to be pushed on the heap at once since the structure arguments are not uniőed and accessed consecutively. Therefore the write structure pushes the functor on the heap and increments H by the arity of the structure. To access structure arguments, both in read and write mode the S register is used. The read and write mode instruction sequences are very similar except that the former contains the read and the latter the write variants of the uniőcation instructions.

Below we present an example how to implement our method in the WAM. Note that for bytecode emulators these instructions are too low-level, several instructions could be merged into one to avoid the dispatching overhead. For native code, on the other hand, the generated code must be postprocessed to remove redundant operations.

read down X_i This instruction is generated before a compound term which is not the last argument. A reference to the next argument is stored into the temporary. X_i = S + 1;

write down X_i Like read down, but the mode write is marked into the tag of the temporary.

X_i = tag struct(S + 1);

read structure F A substructure in read mode. If the term pointed to by S is a structure pointer and the functor of the structure is equal to F, S is set to point to its őrst argument, else a failure occurs.

if (tag(*S) != tag struct || S = val(*S), *S++ != F) Fail;

write structure F A substructure in write mode. A pointer to a new structure skeleton with functor F is stored in the location pointed to by S and S is set to point to the őrst argument of the new structure skeleton.

*S = tag struct(H); H++ = F; S = H;

H += arity(F);

read up X_i This instruction is generated at the end of a structure which is itself a (not last) argument of a compound term. The S pointer is restored from the temporary. S = X_i ;

write up X_i This instruction is generated at the end of a structure in write mode which is not the last argument. The S pointer is restored from the temporary. S = val(X_i);

read list Corresponds to read structure for lists.

if (tag(*S) == tag list) S = val(*S);

else

Fail;

write list Corresponds to write structure for lists.

*S = tag list(H); S = H; H += 2;

read test LabW This instruction precedes a substructure in read mode. S is dereferenced. If the result of dereferencing is a free variable, it is trailed if necessary and a jump to LabW is executed.

S = Deref(S);

if (tag(S) == tag ref) { trail(S);

P = LabW; } write test X_i, LabR This instruction is generated after a write up instruction. If the tag of X_i is not write, a jump to LabR is executed.

if (tag(X_i ) != tag struct) P = LabR;

read constant C A constant argument of a compound term in read mode. The contents of S is dereferenced and uniőed with C. After a successful uniőcation S is incremented.

if (Unify(*Deref(S), C))

S++; else

Fail;

write constant C A constant argument of a compound term in write mode. C is stored into the location pointed to by S and S is incremented. *S++ = C;

3.1.1 Compilation Example

Note that the WAM uses less abstract instructions, however they are more complicated than ours. When expanded, the WAM code is always longer.

f(g(X), b)

Our Code WAM Code

get structure A_i, LabR get structure A_i write down X_1 unify variable X_1 W1: write structureg/1 unify constant b write variable Y_1 get structure g/1, X_1 write up X_1 unify variable Y_1 write test X_1, R1 write constant b branch Ends LabR: read down X_1

read test W1 read structureg/1 read variable Y_1 read up X_1 R1: read constant b Ends:

As we can see, the code generated for read and write mode is very similar, temporary variables are the same in both sequences and the correspondence of labels for branches between the two sequences is straightforward.

3.2 Body Arguments

Unlike head arguments, the compound terms to be constructed in the clause body are known at compile time, they are going to be pushed onto consecutive heap locations. Therefore, when the compound term is created top-down, in a breadth-őrst manner, one pointer is completely sufőcient to build it, no temporary variables are necessary.

The instructions work as follows: the generation starts with the topmost structure, őrst the functor is pushed at the heap top using the H pointer, then the structure arguments are pushed, followed by the next substructure etc. Whenever a compound subargument is encountered, a structure pointer to it is pushed on the heap and then the next argument is built. Thus each new structure is put at the end of a queue and built when it is at its beginning.

put structure F, A_i This instruction corresponds to the beginning of a compound argument of a body subgoal. A functor is pushed on the heap, a structure pointer to it is stored into A_i . A_i = tag struct(H); *H++ = F;

push constant C The constant C is pushed on the heap. Note that C can also be a functor of a compound term. *H++ = C;

push structure Offset This instruction corresponds to a compound subargument. A structure pointer to H + Offset is pushed on the heap top.

*H = tag struct(H + Offset); H++;

push list Offset This instruction corresponds to a compound subargument which is a list.

*H = tag list(H + Offset); H++;

An example of a body goal compilation:

test :- do(parse(s(np, vp), [birds, fly], [])).

Our Code WAM Code (from [7])

put structure parse/3, A_1 put structure s/2, X_2 push structure 3 unify constant np push list 5 unify constant vp push constant [] put list X_4 push constant s/2 unify constant fly push constant np unify nil push constant vp put list X_3 push constant birds unify constant birds push list 1 unify value X_4 push constant fly put structure parse/3, A_1 push constant [] unify value X_2 execute do/1 unify value X_3 unify nil execute do/1

It can be seen that to achieve the same effect the WAM needs 3 temporary variables and more code than our method.

4 Optimizations

The above presented methods are the basic schemata only, there are many optimizations to be applied on them.

fl In the sequence of up, down and test abstract instructions in the

generated native code usually several machine instructions can be omitted, which is easy to do in a peephole optimizer.

fl The unify local value WAM instruction has only its write counterpart,

in the read sequence the normal read value instruction is used.

fl When the head structure contains void variables, it is often possible to

omit part of the read sequence, provided that no nonvoid argument follows the void one(s). For example, the term

[ j ]_{can be simply compiled into}

get list A_i , Ends write void 2 Ends:

fl Ground compound terms in the clause body should not be compiled into

a sequence of instructions - even with a structure copying method they can be shared. The compound term can be created once at compile time and in the clause body the put constant instruction just puts a corresponding structure or list pointer to it.

Similarly, ground compound terms in the head should be constructed at compile time and treated like constants in the write sequence, e.g. f(g(a), g(b)) is compiled into

get structure f/2, A_i , LR write constant < g(a) > write constant < g(b) > branch Ends LR: read down X_1 . . . Ends:

fl If the compound argument is the indexed one, a branch directly to the

read sequence can be executed:

append([], L, L).

append([XjL1], L2, [XjL3]) :- append(L1, L2, L3).

Our Code WAM Code

La: switch on typeA_1, list:Ll, nil:Ln, default:fail switch on typeA_1, list:Ll, nil:Ln, default:fail Lv: get nil A_1 Lv: get nil A_1 Ln: get value A_2, A_3 Ln: get value A_2, A_3 proceed proceed

get list A_1, Ll Ll: get list A_1 write variable X_4 unify variable X_4 write variable A_1 unify variable A_1 branch E1 get list A_3 Ll: read variable X_4 unify value X_4 read variable A_1 unify variable A_3 E1: get list A_3, Lr execute append/3 write value X_4 write variable A_3 branch La Lr: read value X_4

read variable A_3 branch La

fl When more information is available about the instantiation of the caller

arguments, e.g. from abstract interpretation, the compiler can omit certain instructions and regroup the others, so that a simple peephole optimization yields optimal native code. For example, if the clause p(g(f(X), h(X))).

is always called with an instantiated argument, but the arguments of g/2 are known to be unbound, a mixture of read and write sequence can be generated, the compiler can replace some of the read instructions by write ones:

g(f(X), h(X))

Default Mixture

get structure g/2, A_i, LR get structure g/2, A_i , L1 write down X_1 L1: read down X_1 W1: write structuref/1 read test L_2 write variable X_2 write structure f/1 write up X_1 write variable X_2 write test X_1, R1 read up X_1 W2: write structureh/1 read test X_1, L_3 write value X_2 L3: write structure h/1 branch Ends write value X_2 LR: read down X_1

read test W1 read structure f/1 read value X_2 read up X_1

R1: read test W2

read structure h/1 read value X_2 Ends:

Note also that the depth-őrst approach is even more ŕexible - the compiler might decide to compile e.g. the last argument depth-őrst and the other ones breadth-őrst etc.

5 Discussion

Optimizing the uniőcation of compound terms in Prolog is a challenging task. One of the problems of Prolog efőciency is exactly this - inefőcient handling of structures. Often a compound term is created before calling a goal only to decompose it in the called procedure, which is a severe overhead. The structure sharing systems have an advantage here, because the structures do not have to be explicitly created. On the other hand, structure sharing has other disadvantages like poor locality of references and creating long reference chains, so that the only way currently being followed is to optimize the structure copying mechanism.

Despite the problems in the WAM handling of compound terms, there has been very little work published about alternative approaches^1. Current emulated systems seem to use the WAM code since it is compact, but the emulator contains separate versions of the unify instructions for read and write mode. The mode ŕag is not used because each concerned instruction knows to which version of the next instruction it should continue.

Native code systems often use a scheme similar to [6], where explicit read and write sequences are generated, e.g. for f(g(a), g(b))

get structure f/2, A_i , LR write variable X_1 write variable X_2 branch S1 LR: read variable X_1 read variable X_2 S1: get structure g/1, X_1

: : :

In this way the use of the mode ŕag is not necessary, however nested structures are still uniőed separately from the main term and so all the problems with the write mode remain. If the last argument is compound, it can be uniőed directly, without a write variable instruction and so the write mode can be propagated to it. Propagating the write mode to other substructures is obviously difőcult with breadth-first traversal, but it is a trivial task when the depth-first approach is used.

Turk [6] describes an approach where the write mode is propagated using a branch into the middle of the get instruction, to omit dereferencing and testing. The write mode can thus be directly propagated only to the őrst compound

1Our work presented in this paper is based on [2].

argument, its őrst compound subargument etc. The following arguments have to push a free variable on the heap and the get instruction has to dereference and test it like in the basic WAM. In our method the number of branches depends on the actual data, how often it is necessary to switch between the two sequences; the write mode is propagated everywhere.

Recently, Van Roy [5] described a lower-level abstract language for the uniőcation which generates separate sequences for the read and write mode. The write mode sequences use a breadth-őrst approach similar to ours for body terms, but the whole skeleton is pushed at once, so that some of the push instructions need two offsets as arguments, which is less efőcient for bytecode and for some machines. The read mode sequences use depth-őrst traversal, but the code is not shared between the two sequences, except for the last argument, and so the write mode instructions appear also inside the read mode sequence. In this way, Van Roy’s write mode sequence does not contain any redundant operations, however for the expense that this code cannot be shared. For complex structures the size of the generated code grows exponentially. When we use the push instructions instead of the write ones, except for the last argument, and generate an additional write sequence for every label in read test using the push instructions, we obtain a method which is very similar to the Van Roy’s one.

Our methods have the following advantages over the WAM or the above derivatives:

fl Since we are using the depth-őrst approach, our method for head

uniőcation needs less temporary variables than the WAM for wide or shallow structures.

fl We generate explicit read and write sequences and for every source

item (including parentheses) there is at most one read and one write instruction. The code size is proportional to the term size, no code is duplicated.

fl The write mode is automatically propagated to all head substructures, no

intermediate free variables are pushed on the heap in order to save and restore the mode information. If it is necessary to save the mode information, most often this is done in a register. Our method thus performs less memory accesses than the pure WAM.

fl Our method uses the H and S register in a uniform way both in read

and write instructions. It is therefore possible to jump from one sequence tothe other one, or generate directly mixed code.

fl Our method can be used both for depth-őrst and breadth-őrst structure

traversal. After performing the global analysis of the program, the compiler can generate the appropriate code and omit sequences or parts of them which are not necessary.

fl The code for constructing compound terms in the body is optimal. For

general head terms the length of our code is (after postprocessing) minimal.

The drawbacks of our method concern the following cases:

fl For left-balanced head structures like f(g(h(a), b), c) where each

structure has only one compound argument which is not the last one, our method needs more temporary variables than the WAM. Prolog programmers normally prefer right-balanced structures which can be processed using tail-recursive loops.

fl When a structure is uniőed in read mode and all its arguments are

uniőed in write mode, two branches are executed for each argument. In this case, either breadth-őrst approach can be used, or the compiler can generate a mixture of read and write sequences.

Acknowledgements

The author would like to thank to Dominique Henry de Villeneuve, Alexander Herold, Jacques Noye, Bruno Poterie and Joachim Schimpf for fruitful discussions about this topic and for commenting previous versions of this paper and to the referees of the submitted paper for interesting comments about up-to-date systems. Many of the ideas presented here are probably already part of Prolog folklore and so the author would also like to acknowledge all people that have contributed to their development.

Bibliography

[1] John Gabriel, Tim Lindholm, E. L. Lusk, and R. A. Overbeek. A tutorial on the warren abstract machine for computational logic. Technical Report ANL-84-84, Argonne National Laboratory, 1984.

[2] Micha Meier. The compilation of compound term uniőcation in Prolog.

Internal Report IR-LP/lpp1, ECRC, May 1987.

[3] Micha Meier. Analysis of Prolog procedures for indexing purposes. In ICOT, editor, Proceedings of the International Conference on Fifth Generation Computer Systems, pages 800ś807, Tokyo, November 1988.

[4] Micha Meier, Abderrahmane Aggoun, David Chan, Pierre Dufresne, Reinhard Enders, Dominique Henry de Villeneuve, Alexander Herold, Philip Kay, Bruno Perez, Emmanuel van Rossum, and Joachim Schimpf. SEPIA - an extendible Prolog system. In Proceedings of the 11th World Computer Congress IFIP’89, pages 1127ś1132, San Francisco, August 1989.

[5] Peter Van Roy. An intermediate language to support Prolog’s uniőcation. In Ewing L. Lusk and Ross A. Overbeek, editors, Proceedings of the NACLP’89, pages 1148ś1164, Cleveland, October 1989.

[6] Andrew K. Turk. Compiler optimizations for the WAM. In Third International Conference on Logic Programming, pages 657ś662, London, July 1986.

[7] David H. D. Warren. An abstract Prolog instruction set. Technical Note 309, SRI, October 1983.

A Program to Compile Compound Terms

% Compile a head compound term
head(Term) :-
   functor(Term, F, A),
   compile_args(Term, 1, A, 1, Code,
        [branch(lab(_)), label(LR)|ReadCode]),
   read_seq(Code, ReadCode, []),
   pwrite([get_structure(F/A, reg(i), lab(LR))|Code]).

% Compile a compound subgoal argument
body(Term) :-
   functor(Term, F, A),
   A1 is A + 1,
   compile_body([Term|Cont], Cont, A1, [_|Code], []),
   pwrite([put_structure(F/A, reg(i))|Code]).

% Write sequence for the arguments of a compound term
compile_args(Term, A, A, Reg) -->
   {arg(A, Term, Arg)},
   compile_arg(Arg, Reg, last).
compile_args(Term, I, A, Reg) -->
   {I < A, arg(I, Term, Arg), I1 is I + 1},
   compile_arg(Arg, Reg, notlast),
   compile_args(Term, I1, A, Reg).

% Generate the write sequence for one argument
compile_arg(Struct, Reg, last) -->
   {compound(Struct), functor(Struct, F, A)},
   [label(_), write_structure(F/A)],
   compile_args(Struct, 1, A, Reg).
compile_arg(Struct, Reg, notlast) -->
   {compound(Struct), functor(Struct, F, A), Reg1 is Reg+1},
   [write_down(reg(Reg)), label(_), write_structure(F/A)],
   compile_args(Struct, 1, A, Reg1),
   [write_up(reg(Reg)), write_test(lab(_))].
compile_arg(Const, _, _) -->
   {atomic(Const)},
   [write_constant(Const)].

% Generate the read sequence and fill in the labels
read_seq([branch(lab(L))|_]) -->

   [label(L)].
read_seq([write_down(R)|T]) -->
   [read_down(R)],
   read_seq(T).
read_seq([label(L), write_structure(S)|T]) -->
   [read_test(lab(L)), read_structure(S)],
   read_seq(T).
read_seq([write_up(R), write_test(lab(L))|T]) -->
   [read_up(R), label(L)],
   read_seq(T).
read_seq([write_constant(C)|T]) -->
   [read_constant(C)],
   read_seq(T).

% Compile a queue of body structures
compile_body([], [], _) --> {true}.
compile_body([Struct|Rest], Cont, Off) -->
   {functor(Struct, F, A), Off1 is Off - 1},
   [push_constant(F/A)],
   compile_struct(Struct, 1, A, Off1, NewOff, Cont, NewCont),
   compile_body(Rest, NewCont, NewOff).

% Compile one body structure
compile_struct(Struct, A, A, Off, NewOff, Cont, NewCont) -->
   {arg(A, Struct, Arg)},
   compile_body_arg(Arg, Off, NewOff, Cont, NewCont).
compile_struct(Struct, I, A, Off, NewOff, Cont, NewCont) -->
   {I < A, arg(I, Struct, Arg), I1 is I + 1},
   compile_body_arg(Arg, Off, NO, Cont, NC),
   compile_struct(Struct, I1, A, NO, NewOff, NC, NewCont).

% Compile one argument of a body structure
compile_body_arg(Const, Off, NewOff, C, C) -->
   {atomic(Const), NewOff is Off - 1},
   [push_constant(Const)].
compile_body_arg(Struct, Off, NewOff, [Struct|C], C) -->
   {compound(Struct), functor(Struct, _, A),
       NewOff is Off + A},
   [push_structure(Off)].

% Print the generated code
pwrite([]).
pwrite([label(Lab)|Rest]) :-
   write(Lab),
   write(:),
   pwrite(Rest).
pwrite([Instr|Rest]) :-

   put(9),
   functor(Instr, F, A),
   write(F),
   name(F, LS),
   length(LS, Length),
   tab(20-Length),
   writeargs(Instr, 1, A),
   nl,
   pwrite(Rest).

writeargs(Instr, A, A) :-
   arg(A, Instr, Arg),
   writearg(Arg).
writeargs(Instr, I, A) :-
   I < A,
   arg(I, Instr, Arg),
   writearg(Arg),
   write(’, ’),
   I1 is I + 1,
   writeargs(Instr, I1, A).

writearg(lab(L)) :-
   write(L).
writearg(reg(R)) :-
   write(’X’),
   write(R).
writearg(Arg) :-
   write(Arg).